


***********
** Table S2
***********

* Open data
use "Ireland_Data.dta", replace

* Sample descriptives
tab female
tab agegroup2
tab region2
tab degree
tab pastvote2 if pastvote2 != 8 // Exclude "I do not remember"




***********
** Table S3
***********

* Open data
use "Ireland_Data.dta", replace

* Uncomment to install user-written loevh package
*ssc install loevh, replace

* MSA
loevh fair1 fair2 fair3
	* Crit values generated in R: 0; 0; 0

* Cronbach's alpha
alpha fair1 fair2 fair3


	
	
***********
** Table S4
***********

* Open data
use "Ireland_Data.dta", replace

* Uncomment to install user-written loevh package
*ssc install loevh, replace

* MSA
loevh decacc1 decacc2
	* Crit values generated in R: -12; -11
	
* Cronbach's alpha
alpha decacc1 decacc2




***********
** Table S5
***********

* Open data
use "Ireland_Data.dta", replace

* Fairness
reg fair i.expgroup
	
* Acceptance
reg decacc i.expgroup

	


***********
** Table S6
***********

* Open data
use "Ireland_Data.dta", replace

* Fairness
reg fair i.expgroup
	* Representative mini-public vs small demographic bias
	test 2.expgroup = 3.expgroup
	* Representative mini-public vs small attitudinal bias	
	test 2.expgroup = 5.expgroup
	* Small demographic bias vs small attitudinal bias
	test 3.expgroup = 5.expgroup
	
	* Representative mini-public vs large demographic bias
	test 2.expgroup = 4.expgroup
	* Representative mini-public vs large attitudinal bias	
	test 2.expgroup = 6.expgroup
	* Large demographic bias vs large attitudinal bias
	test 4.expgroup = 6.expgroup
	
	* Small demographic bias vs large demographic bias
	test 3.expgroup = 4.expgroup
	* Small attitudinal bias vs large attitudinal bias
	test 5.expgroup = 6.expgroup
	
	* Small attitudinal bias vs large demographic bias
	test 5.expgroup = 4.expgroup
	* Small demographic bias vs large attitudinal bias
	test 3.expgroup = 6.expgroup	
	
	
* Acceptance
reg decacc i.expgroup
	* Representative mini-public vs small demographic bias
	test 2.expgroup = 3.expgroup
	* Representative mini-public vs small attitudinal bias	
	test 2.expgroup = 5.expgroup
	* Small demographic bias vs small attitudinal bias
	test 3.expgroup = 5.expgroup
	
	* Representative mini-public vs large demographic bias
	test 2.expgroup = 4.expgroup
	* Representative mini-public vs large attitudinal bias	
	test 2.expgroup = 6.expgroup
	* Large demographic bias vs large attitudinal bias
	test 4.expgroup = 6.expgroup
	
	* Small demographic bias vs large demographic bias
	test 3.expgroup = 4.expgroup
	* Small attitudinal bias vs large attitudinal bias
	test 5.expgroup = 6.expgroup
	
	* Small attitudinal bias vs large demographic bias
	test 5.expgroup = 4.expgroup
	* Small demographic bias vs large attitudinal bias
	test 3.expgroup = 6.expgroup
	

	
	

***********
** Table S7
***********

* Open data
use "Ireland_Data.dta", replace

* Seemingly Unrelated Regression
sureg (fair i.expgroup) (decacc i.expgroup)

* Compare coefficients across models
di _b[fair:2.expgroup] - _b[decacc:2.expgroup]
test _b[fair:2.expgroup] = _b[decacc:2.expgroup]
di _b[fair:3.expgroup] - _b[decacc:3.expgroup]
test _b[fair:3.expgroup] = _b[decacc:3.expgroup]
di _b[fair:4.expgroup] - _b[decacc:4.expgroup]
test _b[fair:4.expgroup] = _b[decacc:4.expgroup]
di _b[fair:5.expgroup] - _b[decacc:5.expgroup]
test _b[fair:5.expgroup] = _b[decacc:5.expgroup]
di _b[fair:6.expgroup] - _b[decacc:6.expgroup]
test _b[fair:6.expgroup] = _b[decacc:6.expgroup]
di _b[fair:_cons] - _b[decacc:_cons]
test _b[fair:_cons] = _b[decacc:_cons]	





************
** Figure S1
************

* Open data
use "Ireland_Data.dta", replace

* Variables to save results
gen var = "Procedural Fairness" in 1/6
replace var = "Decision Acceptance" in 7/12
gen treat = _n in 1/6
replace treat = _n-6 in 7/12


* Main results
gen pemain = 0 if treat == 1
gen semain = .
reg fair i.expgroup
forvalues i = 2/6 {
	replace pemain = _b[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
	replace semain = _se[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
}
reg decacc i.expgroup
forvalues i = 2/6 {
	replace pemain = _b[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
	replace semain = _se[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
}


* Covariates
gen pecovar = 0 if treat == 1
gen secovar = .
reg fair i.expgroup age female degree i.pastvote2 poltrust satisfdem
forvalues i = 2/6 {
	replace pecovar = _b[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
	replace secovar = _se[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
}
reg decacc i.expgroup age female degree i.pastvote2 poltrust satisfdem
forvalues i = 2/6 {
	replace pecovar = _b[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
	replace secovar = _se[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
}


* Previously heard of citizens' assembly (88% of respondents)
gen peheard = 0 if treat == 1
gen seheard = .
reg fair i.expgroup if heardofca2 == 1
forvalues i = 2/6 {
	replace peheard = _b[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
	replace seheard = _se[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
}
reg decacc i.expgroup if heardofca2 == 1
forvalues i = 2/6 {
	replace peheard = _b[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
	replace seheard = _se[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
}


* Speeders (as pre-registered)
gen pespe = 0 if treat == 1
gen sespe = .
reg fair i.expgroup if speeder == 0
forvalues i = 2/6 {
	replace pespe = _b[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
	replace sespe = _se[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
}
reg decacc i.expgroup if speeder == 0
forvalues i = 2/6 {
	replace pespe = _b[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
	replace sespe = _se[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
}


* Speeders (1/2 of median time)
	* Pre-registered definition of speeder (1/3 of average time) turned out to be problematic because of a small number of respondents with super-long durations (2% of respondents between 2 and 60 hours)
gen pespealt = 0 if treat == 1
gen sespealt = .
reg fair i.expgroup if durationinseconds > 290.5
forvalues i = 2/6 {
	replace pespealt = _b[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
	replace sespealt = _se[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
}
reg decacc i.expgroup if durationinseconds > 290.5
forvalues i = 2/6 {
	replace pespealt = _b[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
	replace sespealt = _se[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
}

* First 1,200 respondents only
gen pen = 0 if treat == 1
gen sen = .
sort startdate_num
reg fair i.expgroup in 1/1200
forvalues i = 2/6 {
	replace pen = _b[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
	replace sen = _se[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
}
sort startdate_num
reg decacc i.expgroup in 1/1200
forvalues i = 2/6 {
	replace pen = _b[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
	replace sen = _se[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
}

* Everything correctly recalled  (exp. condition, DMP recommendation, and final decision)
gen pefmc = 0 if treat == 1
gen sefmc = .
reg fair i.expgroup if fmc_treat_corr == 1 & fmc_rec_corr == 1 & fmc_final_corr == 1 
forvalues i = 2/6 {
	replace pefmc = _b[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
	replace sefmc = _se[`i'.expgroup] if var == "Procedural Fairness" & treat == `i'
}
reg decacc i.expgroup if fmc_treat_corr == 1 & fmc_rec_corr == 1 & fmc_final_corr == 1 
forvalues i = 2/6 {
	replace pefmc = _b[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
	replace sefmc = _se[`i'.expgroup] if var == "Decision Acceptance" & treat == `i'
}

* Change order
gen y = treat
foreach var of varlist pemain - sefmc {
gen `var'2 = `var'
sum `var' if var == "Procedural Fairness" & treat == 4
replace `var'2 = r(mean) if var == "Procedural Fairness" & y == 5
sum `var' if var == "Procedural Fairness" & treat == 5
replace `var'2 = r(mean) if var == "Procedural Fairness" & y == 4
sum `var' if var == "Decision Acceptance" & treat == 4
replace `var'2 = r(mean) if var == "Decision Acceptance" & y == 5
sum `var' if var == "Decision Acceptance" & treat == 5
replace `var'2 = r(mean) if var == "Decision Acceptance" & y == 4
}

* CIs
foreach a in main covar heard spe spealt n fmc {
gen lower`a'2 = pe`a'2 - (1.96 * se`a'2)
gen upper`a'2 = pe`a'2 + (1.96 * se`a'2)
}

* Y-values
gen ymain = y - 0.3
gen ycovar = y - 0.2
gen yheard = y - 0.1
gen yspe = y 
gen yspealt = y + 0.1
gen yn = y + 0.2
gen yfmc = y + 0.3

* Final preparations
gen var_num = 1 if var == "Procedural Fairness"
replace var_num = 2 if var == "Decision Acceptance"
labmask var_num, values(var)


* Graph
twoway ///
	(scatter ymain pemain2, mcolor(black) msymbol(o) msize(medium)) ///
	(rspike lowermain2 uppermain2 ymain, lcolor(black) horizontal) ///
	(scatter ycovar pecovar2, mcolor(gs12) msymbol(s) msize(medium)) ///
	(rspike lowercovar2 uppercovar2 ycovar, lcolor(gs12) horizontal) ///
	(scatter yheard peheard2, mcolor(gs10) msymbol(t) msize(medium)) ///
	(rspike lowerheard2 upperheard2 yheard, lcolor(gs10) horizontal) ///
	(scatter yspe pespe2, mcolor(gs8) msymbol(d) msize(medium)) ///
	(rspike lowerspe2 upperspe2 yspe, lcolor(gs8) horizontal) ///	
	(scatter yspealt pespealt2, mcolor(gs6) msymbol(o) msize(medium)) ///
	(rspike lowerspealt2 upperspealt2 yspealt, lcolor(gs6) horizontal) ///
	(scatter yn pen2, mcolor(gs4) msymbol(s) msize(medium)) ///
	(rspike lowern2 uppern2 yn, lcolor(gs4) horizontal) ///
	(scatter yfmc pefmc2, mcolor(gs2) msymbol(t) msize(medium)) ///
	(rspike lowerfmc2 upperfmc2 yfmc, lcolor(gs2) horizontal) ///
	, ///
	by(var_num, noixlabel ixtitle graphregion(fcolor(white) lcolor(white)) bgcolor(white) note("", margin(top)) legend(pos(3))) ///
	ytitle("") yscale(noline range(0.5 6.5) reverse) ///	
	ylabel(1 "No mini-public (control)" 2 "Representative mini-public" 3 "Small demographic bias" 4 "Small attitudinal bias" 5 "Large demographic bias" 6 "Large attitudinal bias", angle(horizontal) nogrid) ///
	xtitle("Treatment effect", margin(small))  ///
	xline(0, lwidth(thin) lpattern(solid) lcolor(black) extend) ///
	xlabel(-0.2(0.2)0.4,) xmlabel(-0.2(0.1)0.4,) xscale(noline) ///
	subtitle(, size(medlarge) align(middle) margin(bottom) nobox fcolor(white))  ///
	graphregion(fcolor(white) ifcolor(white) lcolor(white)) plotregion(fcolor(white) lcolor(black)) bgcolor(white) ///
	legend(order(1 3 5 7 9 11 13) ///
	label(1 "Main results") label(3 "Covariates") label(5 "Heard of citizens' assembly") ///
	label(7 "No speeders") label(9 "No speeders (alt)") label(11 "{it:N} = 1,200") ///
	label(13 "Correct recall") cols(1) size(small) keygap(*1) region(lstyle(none) lcolor(white))) ///
	scheme(s2mono) xsize(7) ysize(3.5)
gr_edit .b1title.draw_view.setstyle, style(no)
gr_edit .plotregion1.xaxis1[1].style.editstyle majorstyle(tickstyle(show_labels(yes))) editcopy




***********
** Table S8
***********

* Open data
use "Ireland_Data.dta", replace

* Fairness
reg fair i.expgroup##i.degree
	* Joint significance of interaction term
	testparm 2.expgroup#1.degree 3.expgroup#1.degree 4.expgroup#1.degree 5.expgroup#1.degree 6.expgroup#1.degree
	
* Decision acceptance
reg decacc i.expgroup##i.degree
	* Joint significance of interaction term
	testparm 2.expgroup#1.degree 3.expgroup#1.degree 4.expgroup#1.degree 5.expgroup#1.degree 6.expgroup#1.degree






***********
** Table S9
***********

* Open data
use "Ireland_Data.dta", replace

* Fairness
reg fair i.expgroup##i.pos
	* Joint significance of interaction term
	testparm 2.expgroup#1.pos 3.expgroup#1.pos 4.expgroup#1.pos 5.expgroup#1.pos 6.expgroup#1.pos
	
* Decision acceptance
reg decacc i.expgroup##i.pos
	* Joint significance of interaction term
	testparm 2.expgroup#1.pos 3.expgroup#1.pos 4.expgroup#1.pos 5.expgroup#1.pos 6.expgroup#1.pos

	
	


************
** Figure S2
************

* Open data
use "Ireland_Data.dta", replace	

* Extract point and SE estimates
gen var = "Procedural Fairness" in 1/6
replace var = "Decision Acceptance" in 7/12
gen treat = _n in 1/6
replace treat = _n-6 in 7/12
gen pe_nodegree = 0 if treat == 1
gen se_nodegree = .
gen pe_degree = 0 if treat == 1
gen se_degree = .

reg fair i.expgroup##i.degree
margins expgroup#degree, post coeflegend
forvalues i = 2/6 {
lincom _b[`i'.expgroup#0bn.degree] - _b[1bn.expgroup#0bn.degree]
replace pe_nodegree = `r(estimate)' if var == "Procedural Fairness" & treat == `i'
replace se_nodegree = `r(se)' if var == "Procedural Fairness" & treat == `i'
lincom _b[`i'.expgroup#1bn.degree] - _b[1bn.expgroup#1bn.degree]
replace pe_degree = `r(estimate)' if var == "Procedural Fairness" & treat == `i'
replace se_degree = `r(se)' if var == "Procedural Fairness" & treat == `i'
}

reg decacc i.expgroup##i.degree
margins expgroup#degree, post coeflegend
forvalues i = 2/6 {
lincom _b[`i'.expgroup#0bn.degree] - _b[1bn.expgroup#0bn.degree]
replace pe_nodegree = `r(estimate)' if var == "Decision Acceptance" & treat == `i'
replace se_nodegree = `r(se)' if var == "Decision Acceptance" & treat == `i'
lincom _b[`i'.expgroup#1bn.degree] - _b[1bn.expgroup#1bn.degree]
replace pe_degree = `r(estimate)' if var == "Decision Acceptance" & treat == `i'
replace se_degree = `r(se)' if var == "Decision Acceptance" & treat == `i'
}

* Change order
gen y = treat
foreach var of varlist pe_nodegree - se_degree {
gen `var'2 = `var'
sum `var' if var == "Procedural Fairness" & treat == 4
replace `var'2 = r(mean) if var == "Procedural Fairness" & y == 5
sum `var' if var == "Procedural Fairness" & treat == 5
replace `var'2 = r(mean) if var == "Procedural Fairness" & y == 4
sum `var' if var == "Decision Acceptance" & treat == 4
replace `var'2 = r(mean) if var == "Decision Acceptance" & y == 5
sum `var' if var == "Decision Acceptance" & treat == 5
replace `var'2 = r(mean) if var == "Decision Acceptance" & y == 4
}

* CIs
foreach a in nodegree degree {
gen lower`a'2 = pe_`a'2 - (1.96 * se_`a'2)
gen upper`a'2 = pe_`a'2 + (1.96 * se_`a'2)
}

* Y-values
gen ynodegree = y - 0.15
gen ydegree = y + 0.15

* Numerical results as marker labels
gen mlabel_degree = ""
gen mlabel_nodegree = ""

gen temp1 = pe_nodegree2
gen temp2 = pe_degree2
format %03.2f temp1 temp2
tostring temp1 temp2, replace usedisplayformat force
replace mlabel_degree = temp2 in 2/6
replace mlabel_nodegree = temp1 in 2/6
replace mlabel_degree = temp2 in 8/12
replace mlabel_nodegree = temp1 in 8/12
drop temp1 temp2 

* Final preparations
gen var_num = 1 if var == "Procedural Fairness"
replace var_num = 2 if var == "Decision Acceptance"
labmask var_num, values(var)

* Graph
twoway ///
	(scatter ynodegree pe_nodegree2, mcolor(black) msymbol(o) msize(medium) mlabel(mlabel_nodegree) mlabposition(12) mlabangle(horizontal) mlabsize(small)) ///
	(rspike lowernodegree2 uppernodegree2 ynodegree, lcolor(black) horizontal) ///
	(scatter ydegree pe_degree2, mcolor(gs10) msymbol(t) msize(medium) mlabel(mlabel_degree) mlabposition(6) mlabangle(horizontal) mlabsize(small) mlabcolor(gs10)) ///
	(rspike lowerdegree2 upperdegree2 ydegree, lcolor(gs10) horizontal) ///
	, ///
	by(var_num, noixlabel ixtitle graphregion(fcolor(white) lcolor(white)) bgcolor(white) note("", margin(top)) legend(pos(3))) ///
	ytitle("") yscale(noline range(0.5 6.5) reverse) ///	
	ylabel(1 "No mini-public (control)" 2 "Representative mini-public" 3 "Small demographic bias" 4 "Small attitudinal bias" 5 "Large demographic bias" 6 "Large attitudinal bias", angle(horizontal) nogrid) ///
	xtitle("Treatment effect", margin(small))  ///
	xline(0, lwidth(thin) lpattern(solid) lcolor(black) extend) ///
	xlabel(-0.2(0.2)0.4,) xmlabel(-0.2(0.1)0.4,) xscale(noline) ///
	subtitle(, size(medlarge) align(middle) margin(bottom) nobox fcolor(white))  ///
	graphregion(fcolor(white) ifcolor(white) lcolor(white)) plotregion(fcolor(white) lcolor(black)) bgcolor(white) ///
	legend(order(1 3) ///
	label(1 "No degree") label(3 "Degree") cols(1) size(small) keygap(*1) region(lstyle(none) lcolor(white))) ///
	scheme(s2mono) xsize(7) ysize(3.5)
gr_edit .b1title.draw_view.setstyle, style(no)
gr_edit .plotregion1.xaxis1[1].style.editstyle majorstyle(tickstyle(show_labels(yes))) editcopy





************
** Figure S3
************

* Open data
use "Ireland_Data.dta", replace	

* Extract point and SE estimates
gen var = "Procedural Fairness" in 1/6
replace var = "Decision Acceptance" in 7/12
gen treat = _n in 1/6
replace treat = _n-6 in 7/12
gen pe_against = 0 if treat == 1
gen se_against = .
gen pe_pro = 0 if treat == 1
gen se_pro = .

reg fair i.expgroup##i.pos
margins expgroup#pos, post coeflegend
forvalues i = 2/6 {
lincom _b[`i'.expgroup#0bn.pos] - _b[1bn.expgroup#0bn.pos]
replace pe_against = `r(estimate)' if var == "Procedural Fairness" & treat == `i'
replace se_against = `r(se)' if var == "Procedural Fairness" & treat == `i'
lincom _b[`i'.expgroup#1bn.pos] - _b[1bn.expgroup#1bn.pos]
replace pe_pro = `r(estimate)' if var == "Procedural Fairness" & treat == `i'
replace se_pro = `r(se)' if var == "Procedural Fairness" & treat == `i'
}

reg decacc i.expgroup##i.pos
margins expgroup#pos, post coeflegend
forvalues i = 2/6 {
lincom _b[`i'.expgroup#0bn.pos] - _b[1bn.expgroup#0bn.pos]
replace pe_against = `r(estimate)' if var == "Decision Acceptance" & treat == `i'
replace se_against = `r(se)' if var == "Decision Acceptance" & treat == `i'
lincom _b[`i'.expgroup#1bn.pos] - _b[1bn.expgroup#1bn.pos]
replace pe_pro = `r(estimate)' if var == "Decision Acceptance" & treat == `i'
replace se_pro = `r(se)' if var == "Decision Acceptance" & treat == `i'
}

* Change order
gen y = treat
foreach var of varlist pe_against - se_pro {
gen `var'2 = `var'
sum `var' if var == "Procedural Fairness" & treat == 4
replace `var'2 = r(mean) if var == "Procedural Fairness" & y == 5
sum `var' if var == "Procedural Fairness" & treat == 5
replace `var'2 = r(mean) if var == "Procedural Fairness" & y == 4
sum `var' if var == "Decision Acceptance" & treat == 4
replace `var'2 = r(mean) if var == "Decision Acceptance" & y == 5
sum `var' if var == "Decision Acceptance" & treat == 5
replace `var'2 = r(mean) if var == "Decision Acceptance" & y == 4
}

* CIs
foreach a in against pro {
gen lower`a'2 = pe_`a'2 - (1.96 * se_`a'2)
gen upper`a'2 = pe_`a'2 + (1.96 * se_`a'2)
}

* Y-values
gen yagainst = y - 0.15
gen ypro = y + 0.15

* Numerical results as marker labels
gen mlabel_pro = ""
gen mlabel_against = ""

gen temp1 = pe_against2
gen temp2 = pe_pro2
format %03.2f temp1 temp2
tostring temp1 temp2, replace usedisplayformat force
replace mlabel_pro = temp2 in 2/6
replace mlabel_against = temp1 in 2/6
replace mlabel_pro = temp2 in 8/12
replace mlabel_against = temp1 in 8/12
drop temp1 temp2 

* Final preparations
gen var_num = 1 if var == "Procedural Fairness"
replace var_num = 2 if var == "Decision Acceptance"
labmask var_num, values(var)

* Graph
twoway ///
	(scatter yagainst pe_against2, mcolor(black) msymbol(o) msize(medium) mlabel(mlabel_against) mlabposition(12) mlabangle(horizontal) mlabsize(small)) ///
	(rspike loweragainst2 upperagainst2 yagainst, lcolor(black) horizontal) ///
	(scatter ypro pe_pro2, mcolor(gs10) msymbol(t) msize(medium) mlabel(mlabel_pro) mlabposition(6) mlabangle(horizontal) mlabsize(small) mlabcolor(gs10)) ///
	(rspike lowerpro2 upperpro2 ypro, lcolor(gs10) horizontal) ///
	, ///
	by(var_num, noixlabel ixtitle graphregion(fcolor(white) lcolor(white)) bgcolor(white) note("", margin(top)) legend(pos(3))) ///
	ytitle("") yscale(noline range(0.5 6.5) reverse) ///	
	ylabel(1 "No mini-public (control)" 2 "Representative mini-public" 3 "Small demographic bias" 4 "Small attitudinal bias" 5 "Large demographic bias" 6 "Large attitudinal bias", angle(horizontal) nogrid) ///
	xtitle("Treatment effect", margin(small))  ///
	xline(0, lwidth(thin) lpattern(solid) lcolor(black) extend) ///
	xlabel(-0.2(0.2)0.4,) xmlabel(-0.2(0.1)0.4,) xscale(noline) ///
	subtitle(, size(medlarge) align(middle) margin(bottom) nobox fcolor(white))  ///
	graphregion(fcolor(white) ifcolor(white) lcolor(white)) plotregion(fcolor(white) lcolor(black)) bgcolor(white) ///
	legend(order(1 3) ///
	label(1 "Against UBI") label(3 "Pro UBI") cols(1) size(small) keygap(*1) region(lstyle(none) lcolor(white))) ///
	scheme(s2mono) xsize(7) ysize(3.5)
gr_edit .b1title.draw_view.setstyle, style(no)
gr_edit .plotregion1.xaxis1[1].style.editstyle majorstyle(tickstyle(show_labels(yes))) editcopy
	
